1

Model Description

1.1 Governing equations

The vertical coordinate (eta) is defined by:

$$\eta = \frac{p - p_T}{p_s - p_T}\eta_s\,; \quad \eta_s = \frac{p_{ref}(z_s) - p_T}{p_{ref}(0) - p_T}$$

The horizontal equations of motion in the η system may be expressed by

$$\frac{d\mathbf{v}}{dt} = -\nabla_\eta \Phi - \frac{RT_v}{p}\nabla_\eta p - f\mathbf{k} \times \mathbf{v} + \mathbf{F} \tag{1.1}$$

The thermodynamic energy equation (the first law of thermodynamics)

$$\frac{dT}{dt} = \frac{\omega\alpha}{c_p} + Q; \qquad \omega \equiv dp/dt \tag{1.2}$$

The mass continuity equation in the η system is

$$\frac{\partial}{\partial\eta}\left(\frac{\partial p}{\partial t}\right) + \nabla_\eta \cdot \left(\mathbf{v}\frac{\partial p}{\partial\eta}\right) + \frac{\partial}{\partial\eta}\left(\dot{\eta}\frac{\partial p}{\partial\eta}\right) = 0 \tag{1.3}$$

1.2 DIVHOA equations and code

Subroutine DIVHOA calculates divergence correction *DC*, divergence *DIV* and horizontal pressure advection in the thermodynamic equation (horizontal part of omega-alpha).

$$DIV = DC + \nabla_\eta \cdot (\mathbf{v}\Delta p) \tag{1.4}$$

$$\frac{\partial T}{\partial t} = \cdots + \frac{\alpha}{c_p}\mathbf{v}\cdot\nabla p \tag{1.5}$$

The finite-difference equations take the form

$$DC = -\frac{0.88w\left[(\Delta x)^2_{min}(\Delta x)^2\right]}{4\Delta t\Delta x\Delta y}\left(\overline{\Pi}^x - \overline{\Pi}^y\right)$$

where

$$\Pi = \Delta_{x'}P_{y'} + \Delta_{y'}P_{x'} \quad ; \quad P_{x'} = \Delta_{x'}\overline{\Phi}^\eta + \frac{\overline{RT_v}^{x'}}{\overline{p}^\eta}\Delta_{x'}\overline{p}^\eta \quad ; \quad P_{y'} = \Delta_{y'}\overline{\Phi}^\eta + \frac{\overline{RT_v}^{y'}}{\overline{p}^\eta}\Delta_{-y'}\overline{p}^\eta$$

$$\nabla_\eta \cdot (\mathbf{v}\Delta p) = \frac{1}{2\Delta x\Delta y}\left[\begin{array}{l} \frac{1}{3}\left(\Delta_x\left(u\Delta y\overline{\Delta p}^x\right) + \Delta_y\left(v\Delta x\overline{\Delta p}^y\right)\right) \\ +\frac{2}{3}\left(\Delta_{x'}\left(\overline{u\Delta y + v\Delta x}^{y'}\overline{\Delta p}^{x'}\right) + \Delta_{y'}\left(\overline{-u\Delta y + v\Delta x}^{x'}\overline{\Delta p}^{y'}\right)\right)\end{array}\right]$$

```
!!-------------------------------------------------------------------
!!--------------preparatory calculations---------------------------------
!!-------------------------------------------------------------------
  if (sigma) then
    do j=1,jm
    do i=1,im
      filo(i,j)=fis(i,j)
      pdsl(i,j)=pd(i,j)
    end do
```

```
    end do
  else
    do j=1,jm
    do i=1,im
      filo(i,j)=0.0
      pdsl(i,j)=res(i,j)*pd(i,j)
    end do
    end do
  end if

  do l=1,lm
  do j=1,jm
  do i=1,im
    div(i,j,l)=0.0
    omgalf(i,j,l)=0.0
  end do
  end do
  end do

  do j=1,jm
  do i=1,im
        adpdx(i,j)=0.0
        adpdy(i,j)=0.0
  end do
  end do




!!---------------main vertical integration loop----------------------------
  vert_loop: do l=lm,1,-1
```

$$\Delta p = DPDE \qquad \frac{1}{\Delta p} = \frac{1}{DPDE} = RDPD$$

```
    do j=1,jm
    do i=1,im
      dpde(i,j)=deta(l)*pdsl(i,j)
      rdpd(i,j)=1.0/dpde(i,j)
    end do
    end do
```

$$\overline{\Delta p}^{x} = \frac{1}{2}\left(DPDE\left(i+ivw(j),j\right)+DPDE\left(i+ive(j),j\right)\right) = \frac{1}{2}ADPDX(i,j)$$

```
do j=1,jm
do i=1+ihe(j),im+ihw(j)
  adpdx(i,j)=dpde(i+ivw(j),j)+dpde(i+ive(j),j)
end do
end do
```

$$\overline{\Delta p}^{y} = \frac{1}{2}\left(DPDE(i,j-1)+DPDE(i,j+1)\right) = \frac{1}{2}ADPDY(i,j)$$

```
do j=2,jm-1
do i=1,im
  adpdy(i,j)=dpde(i,j-1)+dpde(i,j+1)
end do
end do
```

$$\overline{p}^{\eta} = APEL \qquad \frac{RT_v}{\overline{p}^{\eta}} = \frac{RT_v}{APEL} = RTOP \qquad \overline{\Phi}^{\eta} = \frac{1}{2}FIM$$

```
do j=1,jm
do i=1,im
  apel(i,j)=pt+aeta(l)*pdsl(i,j)
  rtop(i,j,l)=r*t(i,j,l)*(1.0+0.608*q(i,j,l))/apel(i,j)
  fiupk=filo(i,j)+rtop(i,j,l)*dpde(i,j)
  fim(i,j)=filo(i,j)+fiupk
  filo(i,j)=dfl(l)+htm(i,j,l)*(fiupk-dfl(l))
end do
end do
```

```
!--------------diagonal contributions to pressure gradient force--------
```

$$\overline{\Delta p}^{x'} = \frac{1}{2}\left(DPDE(i+ihe(j),j+1)+DPDE(i,j)\right) = \frac{1}{2}ADPDNE(i,j)$$

$$\Delta_{x'}\overline{\Phi}^{\eta} = \Delta_{x'}\frac{1}{2}FIM = \frac{1}{2}\frac{1}{2}\left[2\left(FIM(i+ihe(j),j+1)-FIM(i,j)\right)\right] = \frac{1}{2}\frac{1}{2}PNE(i,j)$$

$$\frac{\overline{RT_v}^{x'}}{\overline{p}^{\eta}} \Delta_{x'}\overline{p}^{\eta} = \frac{1}{2}\frac{1}{2}[2(RTOP(i+ihe(j),j+1,l)+RTOP(i,j,l))]$$
$$(APEL(i+ihe(j),j+1)-APEL(i,j))$$
$$= \frac{1}{2}\frac{1}{2}CNE(i,j)$$

$$\overline{\Delta p}^{x'}\frac{\overline{RT_v}^{x'}}{\overline{p}^{\eta}} \Delta_{x'}\overline{p}^{\eta} = \frac{1}{2}ADPDNE(i,j)\frac{1}{2}\frac{1}{2}CNE(i,j) = \frac{1}{2}\frac{1}{2}\frac{1}{2}PCNE(i,j)$$

```
do j=1,jm-1
do i=1,im+ivw(j)
  adpdne(i,j)=dpde(i+ihe(j),j+1)+dpde(i,j)
  pne(i,j)=2.0*(fim(i+ihe(j),j+1)-fim(i,j))
  cne(i,j)=2.0*(rtop(i+ihe(j),j+1,l)+rtop(i,j,l)) &
              *(apel(i+ihe(j),j+1)-apel(i,j))
  pcne(i,j)=cne(i,j)*adpdne(i,j)
end do
end do
```

$$\overline{\Delta p}^{y'} = \frac{1}{2}(DPDE(i+ihe(j),j-1)+DPDE(i,j)) = \frac{1}{2}ADPDSE(i,j)$$

$$\Delta_{-y'}\overline{\Phi}^{\eta} = \Delta_{-y'}\frac{1}{2}FIM = \frac{1}{2}\frac{1}{2}[2(FIM(i+ihe(j),j-1)-FIM(i,j))] = \frac{1}{2}\frac{1}{2}PSE(i,j)$$

$$\frac{\overline{RT_v}^{y'}}{\overline{p}^{\eta}} \Delta_{-y'}\overline{p}^{\eta} = \frac{1}{2}\frac{1}{2}[2(RTOP(i+ihe(j),j-1,l)+RTOP(i,j,l))]$$
$$(APEL(i+ihe(j),j-1)-APEL(i,j))$$
$$= \frac{1}{2}\frac{1}{2}CSE(i,j)$$

$$\overline{\Delta p}^{y'}\frac{\overline{RT_v}^{y'}}{\overline{p}^{\eta}} \Delta_{-y'}\overline{p}^{\eta} = \frac{1}{2}ADPDSE(i,j)\frac{1}{2}\frac{1}{2}CSE(i,j) = \frac{1}{2}\frac{1}{2}\frac{1}{2}PCSE(i,j)$$

```
do j=2,jm
do i=1,im+ivw(j)
  adpdse(i,j)=dpde(i+ihe(j),j-1)+dpde(i,j)
  pse(i,j)=2.0*(fim(i+ihe(j),j-1)-fim(i,j))
  cse(i,j)=2.0*(rtop(i+ihe(j),j-1,l)+rtop(i,j,l)) &
              *(apel(i+ihe(j),j-1)-apel(i,j))
  pcse(i,j)=cse(i,j)*adpdse(i,j)
end do
end do
```

```
!--------------continuity equation modification-------------------------
```

$$
\begin{aligned}
\Pi &= \Delta_{x'}\left(\Delta_{x'}\overline{\Phi}^{\eta} + \frac{\overline{RT_v}^{x'}}{\overline{p}^{\eta}}\Delta_{x'}\overline{p}^{\eta}\right) + \Delta_{y'}\left(\Delta_{y'}\overline{\Phi}^{\eta} + \frac{\overline{RT_v}^{y'}}{\overline{p}^{\eta}}\Delta_{-y'}\overline{p}^{\eta}\right) \\
&= \Delta_{x'}\left(\frac{1}{2}\frac{1}{2}PNE + \frac{1}{2}\frac{1}{2}CNE\right) + \Delta_{y'}\left(\frac{1}{2}\frac{1}{2}PSE + \frac{1}{2}\frac{1}{2}CSE\right) \\
&= \frac{1}{2}\frac{1}{2}\begin{pmatrix} PNE(i+ivw(j),j) - PNE(i,j-1) + CNE(i+ivw(j),j) - CNE(i,j-1) \\ +PSE(i+ivw(j),j) - PSE(i,j+1) + CSE(i+ivw(j),j) - CSE(i,j+1) \end{pmatrix} \\
&= \frac{1}{2}\frac{1}{2}PCXC
\end{aligned}
$$

```
do j=2,jm-1
do i=1+ihe(j),im-1
  pcxc(i,j)=vbm3(i,j)*vtm(i,j,l)*(pne(i+ivw(j),j) &
           +cne(i+ivw(j),j)+pse(i+ivw(j),j)+cse(i+ivw(j),j) &
           -pne(i,j-1)-cne(i,j-1)-pse(i,j+1)-cse(i,j+1))
end do
end do
```

```
!----------------------------------------------------------------------
```

$$
\begin{aligned}
DC &= -\frac{0.88w\left[(\Delta x)^2_{min}(\Delta x)^2\right]}{4\Delta t\Delta x\Delta y}\left(\overline{\Pi}^{x} - \overline{\Pi}^{y}\right) \\
&= WPDAR\ \left(PCXC(i+ihe(j),j) + PCXC(i+ihw(j),j) - PCXC(i,j+1) - PCXC(i,j-1)\right)
\end{aligned}
$$

```
    do j=3,jm-2
    do i=2,im-1+ivw(j)
      div(i,j,l)=deta(l)*wpdar(i,j) &
               *(pcxc(i+ihe(j),j)-pcxc(i,j+1) &
                +pcxc(i+ihw(j),j)-pcxc(i,j-1))
    end do
    end do
```

```
!--------------lat & long pressure force components---------------------
    do j=2,jm-1
    do i=1+ihe(j),im+ihw(j)
      dcnek=cne(i+ivw(j),j)+cne(i,j-1)
      dcsek=cse(i+ivw(j),j)+cse(i,j+1)
      pcew(i,j)=(dcnek+dcsek)*adpdx(i,j)
      pcns(i,j)=(dcnek-dcsek)*adpdy(i,j)
    end do
    end do
```

```
!--------------lat & lon fluxes & omega-alpha components----------------
```

$$u\Delta y = UDY \quad ; \quad v\Delta x = VDX$$

```
    do j=1,jm
    do i=1,im
      udy(i,j)=dy*u(i,j,l)
      vdx(i,j)=dx(i,j)*v(i,j,l)
    end do
    end do
```

$$u\Delta y\overline{\Delta p}^{x} = UDY(i,j)\frac{1}{2}ADPDX(i,j) = \frac{1}{2}FEW(i,j)$$

$$v\Delta x\overline{\Delta p}^{y} = VDX(i,j)\frac{1}{2}ADPDY(i,j) = \frac{1}{2}FNS(i,j)$$

```
do j=2,jm-1
do i=1+ihe(j),im+ihw(j)
  few(i,j)=udy(i,j)*adpdx(i,j)
  tew(i,j)=udy(i,j)*pcew(i,j)
  fns(i,j)=vdx(i,j)*adpdy(i,j)
  tns(i,j)=vdx(i,j)*pcns(i,j)
end do
end do
```

```
!--------------diagonal fluxes and diagonally averaged wind-------------
```

$$\overline{u\Delta y + v\Delta x}^{y'}\overline{\Delta p}^{x'} = \frac{1}{2}\left(UVD(i+ihe(j),j) + UVD(i,j+1)\right)\frac{1}{2}ADPDNE(i,j)$$
$$= \frac{1}{2}\frac{1}{2}FNE(i,j)$$

```
do j=2,jm-2
do i=1,im-1
  pvnek=(udy(i+ihe(j),j)+vdx(i+ihe(j),j))+(udy(i,j+1)+vdx(i,j+1))
  fne(i,j)=pvnek*adpdne(i,j)
  tne(i,j)=pvnek*pcne(i,j)*2.0
end do
end do
```

$$\overline{-u\Delta y + v\Delta x}^{x'}\overline{\Delta p}^{y'} = \frac{1}{2}\left(UVD(i+ihe(j),j) + UVD(i,j-1)\right)\frac{1}{2}ADPDSE(i,j)$$
$$= \frac{1}{2}\frac{1}{2}FSE(i,j)$$

```
    do j=3,jm-1
    do i=1,im-1
      pvsek=(udy(i+ihe(j),j)-vdx(i+ihe(j),j))+(udy(i,j-1)-vdx(i,j-1))
      fse(i,j)=pvsek*adpdse(i,j)
      tse(i,j)=pvsek*pcse(i,j)*2.0
    end do
    end do
```

$$\begin{aligned}
\nabla_\eta \cdot (\mathbf{v}\Delta p) &= \frac{1}{2\Delta x \Delta y}\left[\begin{array}{l} \frac{1}{3}\left(\Delta_x\left(u\Delta y\overline{\Delta p}^x\right) + \Delta_y\left(v\Delta x\overline{\Delta p}^y\right)\right) \\ +\frac{2}{3}\left(\Delta_{x'}\left(\overline{u\Delta y + v\Delta x}^{y'}\overline{\Delta p}^{x'}\right) + \Delta_{y'}\left(\overline{-u\Delta y + v\Delta x}^{x'}\overline{\Delta p}^{y'}\right)\right)\end{array}\right] \\
&= \frac{1}{2\Delta x \Delta y}\left[\begin{array}{l} \frac{1}{3}\left(\Delta_x\left(\frac{1}{2}FEW\right) + \Delta_y\left(\frac{1}{2}FNS\right)\right) \\ +\frac{2}{3}\left(\Delta_{x'}\left(\frac{1}{2}\frac{1}{2}FNE\right) + \Delta_{y'}\left(\frac{1}{2}\frac{1}{2}FSE\right)\right)\end{array}\right] \\
&= \frac{1}{2\Delta x \Delta y}\frac{1}{3}\frac{1}{2}\left[\begin{array}{l} FEW(i+ihe(j),j) - FEW(i+ihw(j),j) + FNS(i,j+1) - FNS(i,j-1) \\ +FNE(i,j) - FNE(i+ihw(j),j-1) + FSE(i,j) - FSE(i+ihw(j),j+1)\end{array}\right]
\end{aligned}$$

```
!--------------horizontal part of omega-alpha & divergence--------------
    do j=3,jm-2
    do i=2,im-1
      hm(i,j)=htm(i,j,l)*hbm2(i,j)
      omgalf(i,j,l)=(tew(i+ihe(j),j)+tew(i+ihw(j),j)+tns(i,j+1) &
                     +tns(i,j-1)+tne(i,j)+tne(i+ihw(j),j-1)+tse(i,j) &
                     +tse(i+ihw(j),j+1))*rdpd(i,j)*fcp(i,j)*hm(i,j)
      t(i,j,l)=omgalf(i,j,l)+t(i,j,l)
      div(i,j,l)=(((few(i+ihe(j),j)+fns(i,j+1)+fne(i,j)+fse(i,j)) &
          -(few(i+ihw(j),j)+fns(i,j-1)+fne(i+ihw(j),j-1) &
           +fse(i+ihw(j),j+1)))*fdiv(i,j)+div(i,j,l))*hm(i,j)
    end do
    end do




  end do vert_loop


  end subroutine divhoa
```

1.3 PGCOR equations and code

$$\frac{\partial u}{\partial t} = \cdots - \frac{\partial \Phi}{\partial x} - \frac{RT_v}{p}\frac{\partial p}{\partial x} - f_c v \tag{1.6}$$

$$\frac{\partial v}{\partial t} = \cdots - \frac{\partial \Phi}{\partial y} - \frac{RT_v}{p}\frac{\partial p}{\partial y} + f_c u \tag{1.7}$$

$$\frac{\partial u}{\partial t} = \ldots + PGF_x + CF_x \quad ; \qquad \frac{\partial v}{\partial t} = \ldots + PGF_y + CF_y$$

where PGF_x and CF_x are pressure gradient force, and Coriolis force in x direction. For pressure gradient force explicit time scheme is used, while for Coriolis force trapesoidal scheme is used.

$$\begin{aligned} u^\tau &= u^{\tau-1} + PGF_x^{\tau-1} + \frac{\Delta t}{2} f_c \left(v^{\tau-1} + v^\tau\right) \\ v^\tau &= v^{\tau-1} + PGF_y^{\tau-1} - \frac{\Delta t}{2} f_c \left(u^{\tau-1} + u^\tau\right) \end{aligned}$$

This set of equations is solved for u^τ and v^τ.

$$\begin{aligned} u^\tau &= u^{\tau-1} + PGF_x^{\tau-1} + F \left(v^{\tau-1} + v^\tau\right) \\ v^\tau &= v^{\tau-1} + PGF_y^{\tau-1} - F \left(u^{\tau-1} + u^\tau\right) \end{aligned}$$

$$\begin{aligned} u^\tau &= UP + Fv^\tau & \qquad u^\tau &= (FVP + UP) / \left(1 + F^2\right) \\ v^\tau &= VP - Fu^\tau & \qquad v^\tau &= VP - Fu^\tau \end{aligned}$$

where $UP = u^{\tau-1} + PGF_x^{\tau-1} + Fv^{\tau-1}$ and $VP = v^{\tau-1} + PGF_y^{\tau-1} - Fu^{\tau-1}$.

$$PGF_x^{\tau-1} = -\frac{\Delta t}{2\Delta x}\left[\begin{array}{r} \frac{1}{3}\left(\overline{\Delta_{x'}\overline{\Phi}^{\eta}}^{y'} + \overline{\overline{\frac{RT_v}{\overline{p}^{\eta}}}^{x'} \Delta_{x'}\overline{p}^{\eta}}^{y'} + \overline{\Delta_{-y'}\overline{\Phi}^{\eta}}^{x'} + \overline{\overline{\frac{RT_v}{\overline{p}^{\eta}}}^{y'} \Delta_{-y'}\overline{p}^{\eta}}^{x'}\right) \\ + \frac{2}{3\overline{\Delta p}^{x}}\left(\overline{\overline{\Delta p}^{x'} \Delta_{x'}\overline{\Phi}^{\eta}}^{y'} + \overline{\overline{\Delta p}^{x'} \overline{\frac{RT_v}{\overline{p}^{\eta}}}^{x'} \Delta_{x'}\overline{p}^{\eta}}^{y'} + \overline{\overline{\Delta p}^{y'} \Delta_{-y'}\overline{\Phi}^{\eta}}^{x'} + \overline{\overline{\Delta p}^{y'} \overline{\frac{RT_v}{\overline{p}^{\eta}}}^{y'} \Delta_{-y'}\overline{p}^{\eta}}^{x'}\right) \end{array}\right]^{\tau-1}$$

$$PGF_y^{\tau-1} = -\frac{\Delta t}{2\Delta y}\left[\begin{array}{r} \frac{1}{3}\left(\overline{\Delta_{x'}\overline{\Phi}^{\eta}}^{y'} + \overline{\overline{\frac{RT_v}{\overline{p}^{\eta}}}^{x'}\Delta_{x'}\overline{p}^{\eta}}^{y'} - \overline{\Delta_{-y'}\overline{\Phi}^{\eta}}^{x'} - \overline{\overline{\frac{RT_v}{\overline{p}^{\eta}}}^{y'}\Delta_{-y'}\overline{p}^{\eta}}^{x'}\right) \\ +\frac{2}{3\overline{\Delta p}^{y}}\left(\overline{\overline{\Delta p}^{x'}\Delta_{x'}\overline{\Phi}^{\eta}}^{y'} + \overline{\overline{\Delta p}^{x'}\overline{\frac{RT_v}{\overline{p}^{\eta}}}^{x'}\Delta_{x'}\overline{p}^{\eta}}^{y'} - \overline{\overline{\Delta p}^{y'}\Delta_{-y'}\overline{\Phi}^{\eta}}^{x'} - \overline{\overline{\Delta p}^{y'}\overline{\frac{RT_v}{\overline{p}^{\eta}}}^{y'}\Delta_{-y'}\overline{p}^{\eta}}^{x'}\right)\end{array}\right]^{\tau-1}$$

$$\Delta p = DPDE \qquad \frac{1}{\Delta p} = \frac{1}{DPDE} = RDPD$$

```
do j=1,jm
do i=1,im
   dpde(i,j)=pdsl(i,j)*deta(l)
   rdpd(i,j)=1.0/dpde(i,j)
end do
end do
```

$$\overline{\Delta p}^{x} = \frac{1}{2}\left(DPDE\left(i+ivw(j),j\right)+DPDE\left(i+ive(j),j\right)\right) = \frac{1}{2}ADPDX(i,j)$$

$$\frac{1}{\overline{\Delta p}^{x}} = \frac{1}{ADPDX(i,j)} = RDPDX(i,j)$$

$$\overline{\Delta p}^{y} = \frac{1}{2}\left(DPDE(i,j-1)+DPDE(i,j+1)\right) = \frac{1}{2}ADPDY(i,j)$$

$$\frac{1}{\overline{\Delta p}^{y}} = \frac{1}{ADPDY(i,j)} = RDPDY(i,j)$$

```
do j=3,jm-2
do i=2,im-1
  adpdx(i,j)=dpde(i+ivw(j),j)+dpde(i+ive(j),j)
  adpdy(i,j)=dpde(i,j-1)+dpde(i,j+1)
  rdpdx(i,j)=1.0/adpdx(i,j)
  rdpdy(i,j)=1.0/adpdy(i,j)
end do
end do
```

$$\overline{p}^{\eta} = APEL \qquad \frac{RT_v}{\overline{p}^{\eta}} = \frac{RT_v}{APEL} = RTOP \qquad \overline{\Phi}^{\eta} = \frac{1}{2}FIM$$

```
do j=1,jm
do i=1,im
  apel(i,j)=pt+aeta(l)*pdsl(i,j)
  rtop(i,j,l)=r*t(i,j,l)*(1.0+0.608*q(i,j,l))/apel(i,j)
  fiupk=filo(i,j)+rtop(i,j,l)*dpde(i,j)
  fim(i,j)=filo(i,j)+fiupk
  filo(i,j)=dfl(l)+htm(i,j,l)*(fiupk-dfl(l))
end do
end do
```

```
!--------------diagonal contributions to pressure gradient force--------
```

$$\overline{\Delta p}^{x'} = \frac{1}{2}\left(DPDE(i+ihe(j), j+1) + DPDE(i,j)\right) = \frac{1}{2}ADPDNE(i,j)$$

$$\Delta_{x'}\overline{\Phi}^{\eta} = \Delta_{x'}\frac{1}{2}FIM = \frac{1}{2}\frac{1}{2}\left[2\left(FIM(i+ihe(j), j+1) - FIM(i,j)\right)\right] = \frac{1}{2}\frac{1}{2}PNE(i,j)$$

$$\overline{\Delta p}^{x'}\Delta_{x'}\overline{\Phi}^{\eta} = \frac{1}{2}ADPDNE(i,j)\frac{1}{2}\frac{1}{2}PNE(i,j) = \frac{1}{2}\frac{1}{2}\frac{1}{2}PPNE(i,j)$$

$$\overline{\frac{RT_v}{\overline{p}^{\eta}}}^{x'}\Delta_{x'}\overline{p}^{\eta} = \frac{1}{2}\frac{1}{2}\left[2\left(RTOP(i+ihe(j), j+1, l) + RTOP(i,j,l)\right)\right]$$
$$\left(APEL(i+ihe(j), j+1) - APEL(i,j)\right)$$
$$= \frac{1}{2}\frac{1}{2}CNE(i,j)$$

$$\overline{\Delta p}^{x'}\overline{\frac{RT_v}{\overline{p}^{\eta}}}^{x'}\Delta_{x'}\overline{p}^{\eta} = \frac{1}{2}ADPDNE(i,j)\frac{1}{2}\frac{1}{2}CNE(i,j) = \frac{1}{2}\frac{1}{2}\frac{1}{2}PCNE(i,j)$$

```
do j=1,jm-1
do i=1,im+ivw(j)
  adpdne(i,j)=dpde(i+ihe(j),j+1)+dpde(i,j)
  pne(i,j)=2.0*(fim(i+ihe(j),j+1)-fim(i,j))
  ppne(i,j)=pne(i,j)*adpdne(i,j)
  cne(i,j)=2.0*(rtop(i+ihe(j),j+1,l)+rtop(i,j,l)) &
              *(apel(i+ihe(j),j+1)-apel(i,j))
  pcne(i,j)=cne(i,j)*adpdne(i,j)
end do
end do
```

$$\overline{\Delta p}^{y'} = \frac{1}{2}\left(DPDE(i+ihe(j),j-1)+DPDE(i,j)\right) = \frac{1}{2}ADPDSE(i,j)$$

$$\Delta_{-y'}\overline{\Phi}^{\eta} = \Delta_{-y'}\frac{1}{2}FIM = \frac{1}{2}\frac{1}{2}\left[2\left(FIM(i+ihe(j),j-1)-FIM(i,j)\right)\right] = \frac{1}{2}\frac{1}{2}PSE(i,j)$$

$$\overline{\Delta p}^{y'}\Delta_{-y'}\overline{\Phi}^{\eta} = \frac{1}{2}ADPDSE(i,j)\frac{1}{2}\frac{1}{2}PSE(i,j) = \frac{1}{2}\frac{1}{2}\frac{1}{2}PPSE(i,j)$$

$$\begin{aligned}\overline{\frac{RT_v}{\overline{p}^{\eta}}}^{y'}\Delta_{-y'}\overline{p}^{\eta} &= \frac{1}{2}\frac{1}{2}\left[2\left(RTOP(i+ihe(j),j-1,l)+RTOP(i,j,l)\right)\right]\\ &\quad\left(APEL(i+ihe(j),j-1)-APEL(i,j)\right)\\ &= \frac{1}{2}\frac{1}{2}CSE(i,j)\end{aligned}$$

$$\overline{\Delta p}^{y'}\overline{\frac{RT_v}{\overline{p}^{\eta}}}^{y'}\Delta_{-y'}\overline{p}^{\eta} = \frac{1}{2}ADPDSE(i,j)\frac{1}{2}\frac{1}{2}CSE(i,j) = \frac{1}{2}\frac{1}{2}\frac{1}{2}PCSE(i,j)$$

```
do j=2,jm
do i=1,im+ivw(j)
  adpdse(i,j)=dpde(i+ihe(j),j-1)+dpde(i,j)
  pse(i,j)=2.0*(fim(i+ihe(j),j-1)-fim(i,j))
  ppse(i,j)=pse(i,j)*adpdse(i,j)
  cse(i,j)=2.0*(rtop(i+ihe(j),j-1,l)+rtop(i,j,l)) &
              *(apel(i+ihe(j),j-1)-apel(i,j))
  pcse(i,j)=cse(i,j)*adpdse(i,j)
end do
end do
```

```
!--------------lat & long pressure force components---------------------
```

$$\overline{\Delta_{x'}\overline{\Phi}^{\eta}}^{y'} = \frac{1}{2}\frac{1}{2}\overline{PNE}^{y'} = \frac{1}{2}\frac{1}{2}\frac{1}{2}\left(PNE(i+ivw(j),j)+PNE(i,j-1)\right) = \frac{1}{2}\frac{1}{2}\frac{1}{2}DPNE$$

$$\overline{\Delta_{-y'}\overline{\Phi}^{\eta}}^{x'} = \frac{1}{2}\frac{1}{2}\overline{PSE}^{x'} = \frac{1}{2}\frac{1}{2}\frac{1}{2}\left(PSE(i+ivw(j),j)+PSE(i,j+1)\right) = \frac{1}{2}\frac{1}{2}\frac{1}{2}DPSE$$

$$\overline{\Delta_{x'}\overline{\Phi}^{\eta}}^{y'} + \overline{\Delta_{-y'}\overline{\Phi}^{\eta}}^{x'} = \frac{1}{2}\frac{1}{2}\frac{1}{2}DPNE + \frac{1}{2}\frac{1}{2}\frac{1}{2}DPSE = \frac{1}{2}\frac{1}{2}\frac{1}{2}PEW(i,j)$$

$$\overline{\Delta_{x'}\overline{\Phi}^{\eta}}^{y'} - \overline{\Delta_{-y'}\overline{\Phi}^{\eta}}^{x'} = \frac{1}{2}\frac{1}{2}\frac{1}{2}DPNE - \frac{1}{2}\frac{1}{2}\frac{1}{2}DPSE = \frac{1}{2}\frac{1}{2}\frac{1}{2}PNS(i,j)$$

$$\overline{\frac{\overline{RT_v}^{x'}}{\overline{p}^{\eta}}\Delta_{x'}\overline{p}^{\eta}}^{y'} = \frac{1}{2}\frac{1}{2}\overline{CNE}^{y'} = \frac{1}{2}\frac{1}{2}\frac{1}{2}\left(CNE(i+ivw(j),j)+CNE(i,j-1)\right) = \frac{1}{2}\frac{1}{2}\frac{1}{2}DCNE$$

$$\overline{\frac{\overline{RT_v}^{y'}}{\overline{p}^{\eta}}\Delta_{-y'}\overline{p}^{\eta}}^{x'} = \frac{1}{2}\frac{1}{2}\overline{CSE}^{x'} = \frac{1}{2}\frac{1}{2}\frac{1}{2}\left(CSE(i+ivw(j),j)+CSE(i,j+1)\right) = \frac{1}{2}\frac{1}{2}\frac{1}{2}DCSE$$

$$\overline{\frac{\overline{RT_v}^{x'}}{\overline{p}^{\eta}}\Delta_{x'}\overline{p}^{\eta}}^{y'} + \overline{\frac{\overline{RT_v}^{x'}}{\overline{p}^{\eta}}\Delta_{-y'}\overline{p}^{\eta}}^{x'} = \frac{1}{2}\frac{1}{2}\frac{1}{2}DCNE + \frac{1}{2}\frac{1}{2}\frac{1}{2}DCSE = \frac{1}{2}\frac{1}{2}\frac{1}{2}PCEW(i,j)$$

$$\overline{\frac{\overline{RT_v}^{x'}}{\overline{p}^{\eta}}\Delta_{x'}\overline{p}^{\eta}}^{y'} - \overline{\frac{\overline{RT_v}^{x'}}{\overline{p}^{\eta}}\Delta_{-y'}\overline{p}^{\eta}}^{x'} = \frac{1}{2}\frac{1}{2}\frac{1}{2}DCNE - \frac{1}{2}\frac{1}{2}\frac{1}{2}DCSE = \frac{1}{2}\frac{1}{2}\frac{1}{2}PCNS(i,j)$$

```
do j=3,jm-2
do i=2,im-1
  dpne=pne(i+ivw(j),j)+pne(i,j-1)
  dpse=pse(i+ivw(j),j)+pse(i,j+1)
  pew(i,j)=dpne+dpsE
  pns(i,j)=dpne-dpse
  dcne=cne(i+ivw(j),j)+cne(i,j-1)
  dcse=cse(i+ivw(j),j)+cse(i,j+1)
  pcew(i,j)=dcne+dcse
  pcns(i,j)=dcne-dcse
end do
end do
```

$$PGF_x^{\tau-1} = -\frac{\Delta t}{2\Delta x}\left[\begin{array}{r} \frac{1}{3}\left(\frac{1}{2}\frac{1}{2}\frac{1}{2}PCEW + \frac{1}{2}\frac{1}{2}\frac{1}{2}PCNS\right) \\ +\frac{2}{3\overline{\Delta p}^x}\left(\frac{1}{2}\frac{1}{2}\frac{1}{2}\overline{PPNE}^{y'} + \frac{1}{2}\frac{1}{2}\frac{1}{2}\overline{PCNE}^{y'} + \frac{1}{2}\frac{1}{2}\frac{1}{2}\overline{PPSE}^{x'} + \frac{1}{2}\frac{1}{2}\frac{1}{2}\overline{PCSE}^{x'}\right) \end{array}\right]^{\tau-1}$$

```
!---------------update u and v (coriolis & pgf)--------------------------
```

```
do j=3,jm-2
do i=2,im-1
  dpfnek=((ppne(i+ivw(j),j)+ppne(i,j-1)) &
         +(pcne(i+ivw(j),j)+pcne(i,j-1)))*2.0
  dpfsek=((ppse(i+ivw(j),j)+ppse(i,j+1)) &
         +(pcse(i+ivw(j),j)+pcse(i,j+1)))*2.0
  dpfew(i,j)=dpfnek+dpfsek
  dpfns(i,j)=dpfnek-dpfsek
end do
end do

do j=3,jm-2
do i=2,im-1
  f0k=u(i,j,l)*curv(i,j)+f(i,j)
  vm(i,j)=vtm(i,j,l)*vbm2(i,j)
  upk=(dpfew(i,j)*rdpdx(i,j)+pcew(i,j)+pew(i,j))*cpgfu(i,j)+f0k*v(i,j,l)+u(i,j,l)
  vpk=(dpfns(i,j)*rdpdy(i,j)+pcns(i,j)+pns(i,j))*cpgfv-f0k*u(i,j,l)+v(i,j,l)
  utk=u(i,j,l)
```

```
    vtk=v(i,j,l)
    u(i,j,l)=((f0k*vpk+upk)/(f0k*f0k+1.0)-u(i,j,l)) &
             *vm(i,j)+u(i,j,l)
    v(i,j,l)=(vpk-f0k*u(i,j,l)-v(i,j,l)) &
             *vm(i,j)+v(i,j,l)
  end do
  end do
```

1.4 HZADV equations and code

1.4.1 Temperature

$$\frac{\partial T}{\partial t} = \ldots \underbrace{-\mathbf{v} \cdot \nabla T}_{ADT}$$

$$T^{\left(\tau-1+n_1\sqrt{2}/2\right)^*} = T^{\tau-1} + \sqrt{2} ADT^{\tau-1}$$

$$T^{\tau-1+n_1} = T^{\tau-1} + 2ADT^{\left(\tau-1+n_1\sqrt{2}/2\right)^*}$$

$$T^{\left(\tau-1+n_1\sqrt{2}/2\right)^*} = T^{\tau-1} - \frac{\sqrt{2}n_1\Delta t}{2\Delta x\Delta y\Delta p}\left[\begin{array}{c}\frac{1}{3}\left(\overline{u\Delta y\overline{\Delta p}^x\Delta_x T}^x + \overline{v\Delta x\overline{\Delta p}^y\Delta_y T}^y\right)\\ +\frac{2}{3}\left(\overline{\overline{u\Delta y+v\Delta x}^{y'}\overline{\Delta p}^{x'}\Delta_{x'}T}^{x'} + \overline{\overline{-u\Delta y+v\Delta x}^{x'}\overline{\Delta p}^{y'}\Delta_{y'}T}^{y'}\right)\end{array}\right]^{\tau-1}$$

$$T^{\tau-1+n_1} = T^{\tau-1} - \frac{n_1\Delta t}{\Delta x\Delta y\Delta p}\left[\begin{array}{c}\frac{1}{3}\left(\overline{u\Delta y\overline{\Delta p}^x\Delta_x T}^x + \overline{v\Delta x\overline{\Delta p}^y\Delta_y T}^y\right)\\ +\frac{2}{3}\left(\overline{\overline{u\Delta y+v\Delta x}^{y'}\overline{\Delta p}^{x'}\Delta_{x'}T}^{x'} + \overline{\overline{-u\Delta y+v\Delta x}^{x'}\overline{\Delta p}^{y'}\Delta_{y'}T}^{y'}\right)\end{array}\right]^{\left(\tau-1+n_1\sqrt{2}/2\right)^*}$$

$$ADT = -\frac{n_1\Delta t}{2\Delta x\Delta y\Delta p}\left[\begin{array}{c}\frac{1}{3}\left(\overline{u\Delta y\overline{\Delta p}^x\Delta_x T}^x + \overline{v\Delta x\overline{\Delta p}^y\Delta_y T}^y\right)\\ +\frac{2}{3}\left(\overline{\overline{u\Delta y+v\Delta x}^{y'}\overline{\Delta p}^{x'}\Delta_{x'}T}^{x'} + \overline{\overline{-u\Delta y+v\Delta x}^{x'}\overline{\Delta p}^{y'}\Delta_{y'}T}^{y'}\right)\end{array}\right]$$

$$\Delta p = DPDE \qquad \frac{1}{\Delta p} = \frac{1}{DPDE} = RDPD$$

```
do j=1,jm
do i=1,im
   dpde(i,j)=pdsl(i,j)*deta(l)
   rdpd(i,j)=1.0/dpde(i,j)
end do
end do
```

$$\overline{\Delta p}^x = \frac{1}{2}\left(DPDE\left(i+ivw(j),j\right)+DPDE\left(i+ive(j),j\right)\right) = \frac{1}{2}ADPDX\left(i,j\right)$$

```
do j=1,jm
do i=1+ihe(j),im+ihw(j)
  adpdx(i,j)=dpde(i+ivw(j),j)+dpde(i+ive(j),j)
  rdpdx(i,j)=1.0/adpdx(i,j)
end do
end do
```

$$\overline{\Delta p}^y = \frac{1}{2}\left(DPDE\left(i,j-1\right)+DPDE\left(i,j+1\right)\right) = \frac{1}{2}ADPDY\left(i,j\right)$$

```
do j=2,jm-1
do i=1,im
  adpdy(i,j)=dpde(i,j-1)+dpde(i,j+1)
  rdpdy(i,j)=1.0/adpdy(i,j)
end do
end do
```

```
!--------------mass fluxes and mass points advection components---------
```

$$u\Delta y = UDY$$

$$v\Delta x = VDX$$

```
do j=1,jm
do i=1,im
  udy(i,j)=ust(i,j)*dy
  vdx(i,j)=vst(i,j)*dx(i,j)
end do
end do
```

$$u\Delta y\overline{\Delta p}^{x} = UDY(i,j)\frac{1}{2}ADPDX(i,j) = \frac{1}{2}FEW(i,j)$$

$$u\Delta y\overline{\Delta p}^{x}\Delta_x T = \frac{1}{2}FEW(i,j)\left(TST(i+ive(j),j) - TST(i+ivw(j),j)\right) = \frac{1}{2}TEW(i,j)$$

```
do j=1,jm
do i=1+ihe(j),im+ihw(j)
  few(i,j)=udy(i,j)*adpdx(i,j)
  tew(i,j)=few(i,j)*(tst(i+ive(j),j)-tst(i+ivw(j),j))
end do
end do
```

$$v\Delta x\overline{\Delta p}^{y} = VDX(i,j)\frac{1}{2}ADPDY(i,j) = \frac{1}{2}FNS(i,j)$$

$$v\Delta x\overline{\Delta p}^{y}\Delta_y T = \frac{1}{2}FNS(i,j)\left(TST(i,j+1) - TST(i,j-1)\right) = \frac{1}{2}TNS(i,j)$$

```
do j=2,jm-1
do i=1,im
  fns(i,j)=vdx(i,j)*adpdy(i,j)
  tns(i,j)=fns(i,j)*(tst(i,j+1)-tst(i,j-1))
end do
end do
```

```
!--------------diagonal fluxes and diagonally averaged wind-------------
```

$$u\Delta y + v\Delta x = UDY(i,j) + VDX(i,j) = UVD(i,j)$$

```
do j=2,jm-1
do i=1,im
  uvd(i,j)=udy(i,j)+vdx(i,j)
end do
end do
```

$$\begin{aligned}
\overline{u\Delta y + v\Delta x}^{y'}\,\overline{\Delta p}^{x'} &= \frac{1}{2}\left(UVD(i+ihe(j),j) + UVD(i,j+1)\right) \\
&\quad \frac{1}{2}\left(DPDE(i,j) + DPDE(i+ihe(j),j+1)\right) \\
&= \frac{1}{2}\frac{1}{2}FNE(i,j)
\end{aligned}$$

$$\overline{u\Delta y + v\Delta x}^{y'}\,\overline{\Delta p}^{x'}\,\Delta_{x'}T = \frac{1}{2}\frac{1}{2}FNE(i,j)\left(TST(i+ihe(j),j+1) - TST(i,j)\right) = \frac{1}{2}\frac{1}{2}TNE(i,j)$$

```
do j=2,jm-1
do i=1,im-1
  fne(i,j)=(uvd(i+ihe(j),j)+uvd(i,j+1)) &
          *(dpde(i,j)+dpde(i+ihe(j),j+1))
  tne(i,j)=fne(i,j)*(tst(i+ihe(j),j+1)-tst(i,j))
end do
end do
```

$$\begin{aligned}
\overline{-u\Delta y + v\Delta x}^{x'}\,\overline{\Delta p}^{y'} &= \frac{1}{2}\left(UVD(i+ihe(j),j) + UVD(i,j-1)\right) \\
&\quad \frac{1}{2}\left(DPDE(i,j) + DPDE(i+ihe(j),j-1)\right) \\
&= \frac{1}{2}\frac{1}{2}FSE(i,j)
\end{aligned}$$

$$\overline{\overline{-u\Delta y + v\Delta x}^{x'}\,\overline{\Delta p}^{y'}}\,\Delta_{y'} T = \frac{1}{2}\frac{1}{2}FSE(i,j)\,(TST(i+ihe(j),j-1) - TST(i,j)) = \frac{1}{2}\frac{1}{2}TSE(i,j)$$

```
do j=2,jm-1
do i=1,im-1
  fse(i,j)=(temp(i+ihe(j),j)+temp(i,j-1))*(dpde(i,j)+dpde(i+ihe(j),j-1))
  tse(i,j)=fse(i,j)*(tst(i+ihe(j),j-1)-tst(i,j))
end do
end do
```

$$ADT = -\frac{n_1\Delta t}{2\Delta x\Delta y\Delta p}\left[\begin{array}{c} \frac{1}{3}\frac{1}{2}\left(\overline{TEW}^{x} + \overline{TNS}^{y}\right) \\ +\frac{2}{3}\frac{1}{2}\frac{1}{2}\left(\overline{TNE}^{x'} + \overline{TSE}^{y'}\right) \end{array}\right]$$

$$ADT = -\frac{n_1\Delta t}{2\Delta x\Delta y\Delta p}\frac{1}{3}\frac{1}{2}\frac{1}{2}\left[\begin{array}{c} (TEW(i+ihw(j),j) + TEW(i+ihe(j),j) + TNS(i,j-1) + TNS(i,j+1)) \\ +(TNE(i+ihw(j),j-1) + TNE(i,j) + TSE(i,j) + TSE(i+ihw(j),j+1)) \end{array}\right]$$

$$-\frac{n_1\Delta t}{2\Delta x\Delta y}\frac{1}{3}\frac{1}{2}\frac{1}{2} = FAD(i,j)$$

```
do j=3,jm-2
do i=2,im-1+ivw(j)
  adt(i,j)=(tew(i+ihw(j),j)+tew(i+ihe(j),j)+tns(i,j-1)+tns(i,j+1) &
           +tne(i+ihw(j),j-1)+tne(i,j)+tse(i,j)+tse(i+ihw(j),j+1)) &
           *rdpd(i,j)*fad(i,j)
end do
end do
```

1.4.2 Wind (velocity components)

$$\frac{\partial u}{\partial t} = \ldots - \mathbf{v}\cdot\nabla u$$

$$\frac{\partial v}{\partial t} = \ldots - \mathbf{v} \cdot \nabla v$$

$$u^{(\tau-1+n_1/2)^*} = u^{\tau-1} - \frac{n_1 \Delta t}{2\Delta x \Delta y \overline{\Delta p}^{x}} \left[\begin{array}{c} \frac{1}{3} \left(\overline{\overline{u\Delta y \overline{\Delta p}^{x}}^{x} \Delta_x u}^{x} + \overline{\overline{v\Delta x \overline{\Delta p}^{y}}^{x} \Delta_y u}^{y} \right) \\ + \frac{2}{3} \left(\overline{\overline{\overline{u\Delta y + v\Delta x}^{y'} \overline{\Delta p}^{x'}}^{x} \Delta_{x'} u}^{x'} + \overline{\overline{\overline{-u\Delta y + v\Delta x}^{x'} \overline{\Delta p}^{y'}}^{y} \Delta_{y'} u}^{y'} \right) \end{array} \right]^{\tau-1}$$

$$v^{(\tau-1+n_1/2)^*} = v^{\tau-1} - \frac{n_1 \Delta t}{2\Delta x \Delta y \overline{\Delta p}^{y}} \left[\begin{array}{c} \frac{1}{3} \left(\overline{\overline{u\Delta y \overline{\Delta p}^{x}}^{x} \Delta_x v}^{x} + \overline{\overline{v\Delta x \overline{\Delta p}^{y}}^{x} \Delta_y v}^{y} \right) \\ + \frac{2}{3} \left(\overline{\overline{\overline{u\Delta y + v\Delta x}^{y'} \overline{\Delta p}^{x'}}^{x} \Delta_{x'} v}^{x'} + \overline{\overline{\overline{-u\Delta y + v\Delta x}^{x'} \overline{\Delta p}^{y'}}^{y} \Delta_{y'} v}^{y'} \right) \end{array} \right]^{\tau-1}$$

1.5 PDTE equations and code

$$\frac{1}{\eta_s}\frac{\partial p_s}{\partial t} + \nabla_\eta \cdot \left(\mathbf{v}\frac{\partial p}{\partial \eta}\right) + \frac{\partial}{\partial \eta}\left(\dot{\eta}\frac{\partial p}{\partial \eta}\right) = 0$$

```
  subroutine pdte()

  use parmeta
  use ctlblk
  use masks
  use dynam
  use vrbls
  use contin

  integer :: i, j, l
  real, dimension(1:im,1:jm) :: pret, rpsl
!!
!!  start subroutine pdte
!!
```

$$\frac{\partial p_s}{\partial t} = -\int_0^{\eta_s} \nabla \cdot \left(\mathbf{v}\frac{\partial p}{\partial \eta}\right) d\eta$$

$$p_s^\tau = p_s^{\tau-1} - \Delta t \sum_{l=1}^{LM} \nabla \cdot (\mathbf{v}_l \Delta p_l)^{\tau-1}$$

$$\sum_{l=1}^{L} \nabla \cdot (\mathbf{v}_l \Delta p_l) = DIV(I,J,L)$$

```
  do l=2,lm
  do j=1,jm-2
  do i=1,im
      div(i,j,l)=div(i,j,l-1)+div(i,j,l)
  end do
  end do
  end do
```

$$-\sum_{l=1}^{LM} \nabla \cdot (\mathbf{v}_l \Delta p_l) = -DIV(I,J,LM) = PSDT(I,J)$$

```
do j=2,jm-1
do i=1,im
  psdt(i,j)=-div(i,j,lm)
  pret(i,j)=psdt(i,j)*res(i,j)
  rpsl(i,j)=1.0/pdsl(i,j)
end do
end do
```

$$\dot{\eta} = -\frac{\eta}{p_s - p_T}\frac{\partial p_s}{\partial t} - \frac{\eta_s}{p_s - p_T}\sum_{l=1}^{LM} \nabla \cdot (\mathbf{v}_l \Delta p_l)$$

```
!--------------computation of etadt------------------------------------
do l=1,lm-1
do j=2,jm-1
do i=1,im
  etadt(i,j,l)=-(pret(i,j)*eta(l+1)+div(i,j,l))*htm(i,j,l+1)*rpsl(i,j)
end do
end do
end do
```

$$\frac{\partial T}{\partial t} = \cdots + \text{vertical part of } \left(\frac{\omega\alpha}{c_p}\right) = \frac{\alpha}{c_p}\left(\frac{\partial p}{\partial t} + \dot{\eta}\frac{\partial p}{\partial \eta}\right)$$

```
!--------------kinetic energy generation terms in t equation------------
do j=3,jm-2
do i=1,im
  omgalf(i,j,1)=omgalf(i,j,1)-div(i,j,1)*rtop(i,j,1)*ef4t
  t(i,j,1)=t(i,j,1)-div(i,j,1)*rtop(i,j,1)*ef4t
end do
end do

do l=2,lm-1
```

```
do j=3,jm-2
do i=1,im
  omgalf(i,j,l)=omgalf(i,j,l)-(div(i,j,l-1)+div(i,j,l))*rtop(i,j,l)*ef4t
  t(i,j,l)=t(i,j,l)-(div(i,j,l-1)+div(i,j,l))*rtop(i,j,l)*ef4t
end do
end do
end do

do j=3,jm-2
do i=1,im
  omgalf(i,j,lm)=omgalf(i,j,lm)+(pret(i,j)-div(i,j,lm-1))*rtop(i,j,lm)*ef4t
  t(i,j,lm)=t(i,j,lm)+(pret(i,j)-div(i,j,lm-1))*rtop(i,j,lm)*ef4t
end do
end do




do l=lm,2,-1
do j=1,jm-2
do i=1,im
    div(i,j,l)=div(i,j,l)-div(i,j,l-1)
end do
end do
end do

end subroutine pdte
```